

```

//stop on this frame
stop();

//create a new loader to load up the imageBox swf file
var myLoader:Loader = new Loader();

//create a new instance of the loader bar movie clip
var loading = new loadingClip();
//position it
loading.x = 100;
loading.y = 150;
//add it to the stage
this.addChild( loading );

//create a URL request that points to the swf we intend to load
var myURL:URLRequest = new URLRequest( "imageBox.swf" );
//add an event for when the loading is complete
myLoader.contentLoaderInfo.addEventListener( Event.COMPLETE, addSwf );
//add an event that fires while the file is still loading
myLoader.contentLoaderInfo.addEventListener( ProgressEvent.PROGRESS, progressHandler );
//begin the load
myLoader.load( myURL );

//this function is called while the file is still loading
function progressHandler( e:ProgressEvent ):void {
    //calculate how far we have loaded so far
    var progressAmount:Number = e.target.bytesLoaded/e.target.bytesTotal;
    //set our loader bar scale to that amount.
    loading.lbar.scaleX = progressAmount;
}

//this function is called when the file is fully loaded
function addSwf( e:Event ):void {

    //we position the newly loaded content
    myLoader.x = 100;
    myLoader.y = 50;

    //then we remove our loading bar
    this.removeChild( loading );

    //next we name this clip for future reference
    myLoader.name = "loadedContent";

    //finally we add it to the stage
    this.addChild( myLoader );

    //and play any animations we have ready
    play();
}

```